Die Friedmann-Gleichungen

FLRW-Metrik

```
q = {t, r, th, ph};
```

```
g = a[t]^2 {{1/a[t]^2, 0, 0, 0}, {0, -1/(1-K r^2), 0, 0},
        {0, 0, -r^2, 0}, {0, 0, 0, -r^2 Sin[th]^2}};
```

```
MatrixForm [g]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -\frac{a[t]^2}{1-K r^2} & 0 & 0 \\ 0 & 0 & -r^2 a[t]^2 & 0 \\ 0 & 0 & 0 & -r^2 a[t]^2 \mathrm{Sin}[th]^2 \end{pmatrix}$$

```
dq = {dt, dr, dth, dph};
```

```
FullSimplify [dq.g.dq]
```

$$dt^2 + a[t]^2 \left(-dth^2 r^2 + \frac{dr^2}{-1+K r^2} - dph^2 r^2 \mathrm{Sin}[th]^2\right)$$

```
gcontra = Inverse[g];
```

Christoffelsymbole:

```
christ= Table[Table[Sum [1/2 gcontra[[ii]][[mi ]] (D[g[[mi ]][[ki]], q[[li]]] +
                 D[g[[mi ]][[li]], q[[ki]]] - D[g[[ki]][[li]], q[[mi ]]]),
        {mi , 1, 4}], {ki, 1, 4}, {li, 1, 4}], {ii, 1, 4}];
```

```
Do[Print[MatrixForm [christ[[ii]]]], {ii, 1, 4}]
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & \frac{a[t]\, a'[t]}{1-K r^2} & 0 & 0 \\ 0 & 0 & r^2 a[t]\, a'[t] & 0 \\ 0 & 0 & 0 & r^2 a[t] \mathrm{Sin}[th]^2 a'[t] \end{pmatrix}$$

$$\begin{pmatrix} 0 & \frac{a'[t]}{a[t]} & 0 & 0 \\ \frac{a'[t]}{a[t]} & \frac{K r}{1-K r^2} & 0 & 0 \\ 0 & 0 & -r\,(1-K r^2) & 0 \\ 0 & 0 & 0 & -r\,(1-K r^2) \mathrm{Sin}[th]^2 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & \frac{a'[t]}{a[t]} & 0 \\ 0 & 0 & \frac{1}{r} & 0 \\ \frac{a'[t]}{a[t]} & \frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & -\mathrm{Cos}[th] \mathrm{Sin}[th] \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 & \frac{a'[t]}{a[t]} \\ 0 & 0 & 0 & \frac{1}{r} \\ 0 & 0 & 0 & \mathrm{Cot}[th] \\ \frac{a'[t]}{a[t]} & \frac{1}{r} & \mathrm{Cot}[th] & 0 \end{pmatrix}$$

Daraus ergibt sich sofort für den Ricci-Tensor (vollständig kovariante Komponenten) und der Ricci-Skalar

```
RicciTen = -FullSimplify[
      Table[Sum [D[christ[[li]][[ii]][[ki]], q[[li]]] - D[christ[[li]][[ii]][[li]],
            q[[ki]]] + Sum [christ[[li]][[ii]][[ki]] christ[[mi ]][[li]][[mi ]] -
              christ[[mi ]][[ii]][[li]] christ[[li]][[ki]][[mi ]], {mi , 1, 4}],
         {li, 1, 4}], {ii, 1, 4}, {ki, 1, 4}]];
MatrixForm [RicciTen]
```

$$\begin{pmatrix} \frac{3\,a''[t]}{a[t]} & 0 & 0 & 0 \\ 0 & -\frac{2\left(K+a'[t]^2\right)+a[t]\,a''[t]}{1-K r^2} & 0 & 0 \\ 0 & 0 & -r^2\left(2\left(K+a'[t]^2\right)+a[t]\,a''[t]\right) & 0 \\ 0 & 0 & 0 & -r^2\,\mathrm{Sin}[th]^2\left(2\left(K+a'[t]^2\right)+a[t]\right. \end{pmatrix}$$

```
RScal = FullSimplify[
    Sum [RicciTen[[al]][[be]] gcontra[[al]][[be]], {al, 1, 4}, {be, 1, 4}]]
```

$$\frac{6\left(K+a'[t]^2+a[t]\,a''[t]\right)}{a[t]^2}$$

Energie-Impuls-Tensor (vollständig kontravariante Komponenten)

```
u = {1, 0, 0, 0};
Tcontra = FullSimplify [Table[-P[t] gcontra[[al]][[be]] +
         (eps[t] + P[t]) u[[al]] u[[be]], {al, 1, 4}, {be, 1, 4}]];
MatrixForm [Tcontra]
```

$$\begin{pmatrix} \mathrm{eps}[t] & 0 & 0 & 0 \\ 0 & \frac{\left(1-K r^2\right) P[t]}{a[t]^2} & 0 & 0 \\ 0 & 0 & \frac{P[t]}{r^2\,a[t]^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[th]^2\,P[t]}{r^2\,a[t]^2} \end{pmatrix}$$

```
T = Sum [(Tcontra.g)[[mu ]][[mu ]], {mu , 1, 4}]
```

$$\mathrm{eps}[t] - 3\,P[t]$$

```
T
```

Einstein - Gleichungen

```
MatrixForm [RicciTen]
```

$$\begin{pmatrix} \frac{3\,a''[t]}{a[t]} & 0 & 0 & 0 \\ 0 & -\frac{2\left(K+a'[t]^2\right)+a[t]\,a''[t]}{1-K r^2} & 0 & 0 \\ 0 & 0 & -r^2\left(2\left(K+a'[t]^2\right)+a[t]\,a''[t]\right) & 0 \\ 0 & 0 & 0 & -r^2\,\mathrm{Sin}[th]^2\left(2\left(K+a'[t]^2\right)+a[t]\right. \end{pmatrix}$$

```
Tcov = g.Tcontra.g;
MatrixForm [Tcov]
```

$$\begin{pmatrix} \mathrm{eps}[t] & 0 & 0 & 0 \\ 0 & \frac{a[t]^2\,P[t]}{1-K r^2} & 0 & 0 \\ 0 & 0 & r^2\,a[t]^2\,P[t] & 0 \\ 0 & 0 & 0 & r^2\,a[t]^2\,P[t]\,\mathrm{Sin}[th]^2 \end{pmatrix}$$

```
MatrixForm [FullSimplify [-ka (Tcov - T / 2 g) + La g]]
```

$$\begin{pmatrix} La - \frac{1}{2} ka (eps[t] + 3 P[t]) & 0 & 0 & \\ 0 & \frac{a[t]^2 (2 La + ka eps[t] - ka P[t])}{-2 + 2 K r^2} & 0 & \\ 0 & 0 & -\frac{1}{2} r^2 a[t]^2 (2 La + ka eps[t] - ka P[t]) & \\ 0 & 0 & 0 & -\frac{1}{2} r \end{pmatrix}$$

Lokale Energieerhaltung

```
Table[Sum [D[Tcontra[[mu ]][[nu]], q[[mu ]]], {mu , 1, 4}] +
    Sum [christ[[nu]][[mu ]][[al]] Tcontra[[mu ]][[al]] +
       christ[[mu ]][[mu ]][[al]] Tcontra[[al]][[nu]],
     {mu , 1, 4}, {al, 1, 4}], {nu, 1, 4}]
```

$$\left\{\frac{3\, eps[t]\, a'[t]}{a[t]} + \frac{3\, P[t]\, a'[t]}{a[t]} + eps'[t], 0, 0, 0\right\}$$

```
Table[Sum [D[gcontra[[mu ]][[nu]], q[[mu ]]], {mu , 1, 4}] +
    Sum [christ[[nu]][[mu ]][[al]] gcontra[[mu ]][[al]] +
       christ[[mu ]][[mu ]][[al]] gcontra[[al]][[nu]],
     {mu , 1, 4}, {al, 1, 4}], {nu, 1, 4}]
```

{0, 0, 0, 0}